



Adafruit TSL2591 High Dynamic Range Digital Light Sensor

Created by lady ada




Last updated on 2019-07-26 07:29:34 PM UTC

Overview




When the future is dazzlingly-bright, this ultra-high-range luminosity sensor will help you measure it. The TSL2591 luminosity sensor is an advanced digital light sensor, ideal for use in a wide range of light situations. Compared to low cost CdS cells, this sensor is more precise, allowing for exact lux calculations and can be configured for different gain/timing ranges to detect light ranges from up to 188uLux up to 88,000 Lux on the fly.

The best part of this sensor is that it **contains both infrared and full spectrum diodes!** That means you can separately measure infrared, full-spectrum or human-visible light. Most sensors can only detect one or the other, which does not accurately represent what human eyes see (since we cannot perceive the IR light that is detected by most photo diodes)




This sensor is much like the TSL2561 but with a wider range (and the interface code is different). This sensor has a massive 600,000,000:1 dynamic range! Unlike the TSL2561 you cannot change the I2C address either, so keep that in mind.



The built in ADC means you can use this with any microcontroller, even if it doesn't have analog inputs. The current draw is extremely low, so its great for low power data-logging systems. about 0.4mA when actively sensing, and less than 5 uA when in power-down mode.

Pinouts

The TSL2591 is a I2C sensor. That means it uses the two I2C data/clock wires available on most microcontrollers, and can share those pins with other sensors as long as they don't have an address collision. For future reference, the I2C address is **0x29** and you *can't* change it!



Power Pins:

- **Vin** - this is the power pin. Since the chip uses 3 VDC, we have included a voltage regulator on board that will take 3-5VDC and safely convert it down. To power the board, give it the same power as the logic level of your microcontroller - e.g. for a 5V micro like Arduino, use 5V
- **3vo** - this is the 3.3V output from the voltage regulator, you can grab up to 100mA from this if you like
- **GND** - common ground for power and logic



(<https://adafru.it/dGy>) I2C Logic pins:

- **SCL** - I2C clock pin, connect to your microcontrollers I2C clock line.
- **SDA** - I2C data pin, connect to your microcontrollers I2C data line.


Other Pins:

- **INT** - this is the INTerrupt pin from the sensor. It can be programmed to do a couple different things by noodling with the i2c registers. For example trigger when a conversion is done, or when the light level has changed a lot, etc. We don't have library support for this pin


Assembly



Prepare the header strip:
Cut the strip to length if necessary. It will be easier to solder if you insert it into a breadboard - **long pins down**






Add the breakout board:
Place the breakout board over the pins so that the short pins poke through the breakout pads



And Solder!
Be sure to solder all pins for reliable electrical contact.

(For tips on soldering, be sure to check out our [Guide to Excellent Soldering](https://adafruit.it/aTk) (<https://adafruit.it/aTk>)).






You're done! Check your solder joints visually and continue onto the next steps

Wiring & Test

You can easily wire this breakout to any microcontroller, we'll be using an Arduino. For another kind of microcontroller, just make sure it has I2C, then port the code - its pretty simple stuff!



(<https://adafru.it/dBn>)


- Connect **Vin** to the power supply, 3-5V is fine. Use the same voltage that the microcontroller logic is based off of. For most Arduinos, that is 5V
- Connect **GND** to common power/data ground
- Connect the **SCL** pin to the I2C clock **SCL** pin on your Arduino. On an UNO & '328 based Arduino, this is also known as **A5**, on a Mega it is also known as **digital 21** and on a Leonardo/Micro, **digital 3**
- Connect the **SDA** pin to the I2C data **SDA** pin on your Arduino. On an UNO & '328 based Arduino, this is also known as **A4**, on a Mega it is also known as **digital 20** and on a Leonardo/Micro, **digital 2**

The TSL2591 has a default I2C address of BOTH **0x29** and **0x28** (which we don't use in code but it definitely is taken) and these cannot be changed!


Install Adafruit_TSL2591 library

To begin reading sensor data, you will need to [install the Adafruit_TSL2591 library \(code on our github repository\) \(https://adafru.it/dGz\)](https://adafru.it/dGz). It is available from the Arduino library manager so we recommend using that.

From the IDE open up the library manager...



And type in `adafruit tsl2591` to locate the library. Click **Install**




We also have a great tutorial on Arduino library installation at:
<http://learn.adafruit.com/adafruit-all-about-arduino-libraries-install-use> (<https://adafru.it/aYM>)

Install Adafruit_Sensor


The TSL2591 library uses the [Adafruit_Sensor support backend](https://adafru.it/aZm) (<https://adafru.it/aZm>) so that readings can be normalized between sensors.

Search the library manager for **Adafruit Unified Sensor** and install that too (you may have to scroll a bit)




Load Demo

Open up **File->Examples->Adafruit_TSL2591->tsl2591** and upload to your Arduino wired up to the sensor



That's it! Now open up the serial terminal window at 9600 speed to begin the test.



Try covering with your hand or shining a lamp onto the sensor to experiment with the light levels!

Library Reference

The **Adafruit_TSL2591** library contains a number of public functions to help you get started with this sensor.

Constructor

To create an instance of the Adafruit_TSL2591 driver, simply declare an appropriate object, along with a 32-bit numeric value to identify this sensor (in case you have several TSL2591s and want to track them separately in a logging system).

```
Adafruit_TSL2591 tsl = Adafruit_TSL2591(2591);
```

Gain and Timing

You can adjust the gain settings and integration time of the sensor to make it more or less sensitive to light, depending on the environment where the sensor is being used.

The gain can be set to one of the following values (though the last value, MAX, has limited use in the real world given the extreme amount of gain applied):

- **TSL2591_GAIN_LOW**: Sets the gain to 1x (bright light)
- **TSL2591_GAIN_MEDIUM**: Sets the gain to 25x (general purpose)
- **TSL2591_GAIN_HIGH**: Sets the gain to 428x (low light)
- **TSL2591_GAIN_MAX**: Sets the gain to 9876x (extremely low light)

Gain can be read or set via the following functions:

- `void setGain(tsl2591Gain_t gain);`
- `tsl2591Gain_t getGain();`

The integration time can be set between 100 and 600ms, and the longer the integration time the more light the sensor is able to integrate, making it more sensitive in low light the longer the integration time. The following values can be used:

- **TSL2591_INTEGRATIONTIME_100MS**
- **TSL2591_INTEGRATIONTIME_200MS**
- **TSL2591_INTEGRATIONTIME_300MS**
- **TSL2591_INTEGRATIONTIME_400MS**
- **TSL2591_INTEGRATIONTIME_500MS**
- **TSL2591_INTEGRATIONTIME_600MS**

The integration time can be read or set via the following functions:

- `void setTiming (tsl2591IntegrationTime_t integration);`
- `tsl2591IntegrationTime_t getTiming();`

An example showing how these functions are used can be seen in the code below:

```

/*****/
/*
  Configures the gain and integration time for the TSL2561
*/
/*****/
void configureSensor(void)
{
  // You can change the gain on the fly, to adapt to brighter/dimmer light situations
  //tsl.setGain(TSL2591_GAIN_LOW);    // 1x gain (bright light)
  tsl.setGain(TSL2591_GAIN_MED);     // 25x gain
  //tsl.setGain(TSL2591_GAIN_HIGH);   // 428x gain

  // Changing the integration time gives you a longer time over which to sense light
  // longer timelines are slower, but are good in very low light situations!
  tsl.setTiming(TSL2591_INTEGRATIONTIME_100MS); // shortest integration time (bright light)
  //tsl.setTiming(TSL2591_INTEGRATIONTIME_200MS);
  //tsl.setTiming(TSL2591_INTEGRATIONTIME_300MS);
  //tsl.setTiming(TSL2591_INTEGRATIONTIME_400MS);
  //tsl.setTiming(TSL2591_INTEGRATIONTIME_500MS);
  //tsl.setTiming(TSL2591_INTEGRATIONTIME_600MS); // longest integration time (dim light)

  /* Display the gain and integration time for reference sake */
  Serial.println("-----");
  Serial.print ("Gain:          ");
  tsl2591Gain_t gain = tsl.getGain();
  switch(gain)
  {
    case TSL2591_GAIN_LOW:
      Serial.println("1x (Low)");
      break;
    case TSL2591_GAIN_MED:
      Serial.println("25x (Medium)");
      break;
    case TSL2591_GAIN_HIGH:
      Serial.println("428x (High)");
      break;
    case TSL2591_GAIN_MAX:
      Serial.println("9876x (Max)");
      break;
  }
  Serial.print ("Timing:          ");
  Serial.print((tsl.getTiming() + 1) * 100, DEC);
  Serial.println(" ms");
  Serial.println("-----");
  Serial.println("");
}

```

Unified Sensor API

The Adafruit_TSL2591 library makes use of the [Adafruit unified sensor framework \(https://adafru.it/dGB\)](https://adafru.it/dGB) to provide sensor data in a standardized format and scale. If you wish to make use of this framework, the two key functions that you need to work with are `getEvent` and `getSensor`, as described below:

```
void getEvent(sensors_event_t*)
```

This function will read a single sample from the sensor and return it in a generic `sensors_event_t` object. To use this function, you simply pass in a `sensors_event_t` reference, which will be populated by the function, and then read the

results, as shown in the following code:

```
/******  
/*  
    Performs a read using the Adafruit Unified Sensor API.  
*/  
/******  
void unifiedSensorAPIRead(void)  
{  
    /* Get a new sensor event */  
    sensors_event_t event;  
    tsl.getEvent(&event);  
  
    /* Display the results (light is measured in lux) */  
    Serial.print("[ "); Serial.print(event.timestamp); Serial.print(" ms ] ");  
    if ((event.light == 0) |  
        (event.light > 4294966000.0) |  
        (event.light < -4294966000.0))  
    {  
        /* If event.light = 0 lux the sensor is probably saturated */  
        /* and no reliable data could be generated! */  
        /* if event.light is +/- 4294967040 there was a float over/underflow */  
        Serial.println("Invalid data (adjust gain or timing)");  
    }  
    else  
    {  
        Serial.print(event.light); Serial.println(" lux");  
    }  
}
```



Note that some checks need to be performed on the sensor data in case the sensor saturated. If saturation happens, please adjust the gain and integration time up or down to change the sensor's sensitivity and output range.

`void getSensor(sensor_t*)`

This function returns some basic information about the sensor, and operates in a similar fashion to `getEvent`. You pass in an empty `sensor_t` reference, which will be populated by this function, and we can then read the results and retrieve some key details about the sensor and driver, as shown in the code below:

```

/*****/
/*
  Displays some basic information on this sensor from the unified
  sensor API sensor_t type (see Adafruit_Sensor for more information)
*/
/*****/
void displaySensorDetails(void)
{
  sensor_t sensor;
  tsl.getSensor(&sensor);
  Serial.println("-----");
  Serial.print ("Sensor:      "); Serial.println(sensor.name);
  Serial.print ("Driver Ver:  "); Serial.println(sensor.version);
  Serial.print ("Unique ID:   "); Serial.println(sensor.sensor_id);
  Serial.print ("Max Value:   "); Serial.print(sensor.max_value); Serial.println(" lux");
  Serial.print ("Min Value:   "); Serial.print(sensor.min_value); Serial.println(" lux");
  Serial.print ("Resolution:  "); Serial.print(sensor.resolution); Serial.println(" lux");
  Serial.println("-----");
  Serial.println("");
  delay(500);
}

```

Raw Data Access API

If you don't wish to use the Unified Sensor API, you can access the raw data for this sensor via the following three functions:

- `uint16_t getLuminosity (uint8_t channel);`
- `uint32_t getFullLuminosity ();`
- `uint32_t calculateLux (uint16_t ch0, uint16_t ch1);`

`getLuminosity` can be used to read either the visible spectrum light sensor, or the infrared light sensor. It will return the raw 16-bit sensor value for the specified channel, as shown in the code below:

```

/*****/
/*
  Shows how to perform a basic read on visible, full spectrum or
  infrared light (returns raw 16-bit ADC values)
*/
/*****/
void simpleRead(void)
{
  // Simple data read example. Just read the infrared, fullspectrum diode
  // or 'visible' (difference between the two) channels.
  // This can take 100-600 milliseconds! Uncomment whichever of the following you want to read
  uint16_t x = tsl.getLuminosity(TSL2591_VISIBLE);
  //uint16_t x = tsl.getLuminosity(TSL2561_FULLSPECTRUM);
  //uint16_t x = tsl.getLuminosity(TSL2561_INFRARED);

  Serial.print("[ "); Serial.print(millis()); Serial.print(" ms ] ");
  Serial.print("Luminosity: ");
  Serial.println(x, DEC);
}

```

`getFullLuminosity` reads both the IR and full spectrum sensors at the same time to allow tighter correlation between the values, and then separates them in SW. The function returns a 32-bit value which needs to be split into two 16-bit

values, as shown in the code below:

```
/*
*****
/*
  Show how to read IR and Full Spectrum at once and convert to lux
*/
*****
void advancedRead(void)
{
  // More advanced data read example. Read 32 bits with top 16 bits IR, bottom 16 bits full spectrum
  // That way you can do whatever math and comparisons you want!
  uint32_t lum = tsl.getFullLuminosity();
  uint16_t ir, full;
  ir = lum >> 16;
  full = lum & 0xFFFF;
  Serial.print("[ "); Serial.print(millis()); Serial.print(" ms ] ");
  Serial.print("IR: "); Serial.print(ir); Serial.print(" ");
  Serial.print("Full: "); Serial.print(full); Serial.print(" ");
  Serial.print("Visible: "); Serial.print(full - ir); Serial.print(" ");
  Serial.print("Lux: "); Serial.println(tsl.calculateLux(full, ir));
}
```

calculateLux can be used to take both the infrared and visible spectrum sensor data and roughly correlate with the equivalent SI lux value, based on a formula from the silicon vendor that takes into account the sensor properties and the integration time and gain settings of the device.

To calculate the lux, simply call **calculateLux(full, ir)**, where 'full' and 'ir' are raw 16-bit values taken from one of the two raw data functions above. See the code sample above for an example of calculating lux.

Arduino Library Docs

[Arduino Library Docs \(https://adafru.it/AvJ\)](https://adafru.it/AvJ)


Python & CircuitPython

It's easy to use the TSL2591 sensor with Python and CircuitPython and the [Adafruit CircuitPython TSL2591 \(https://adafru.it/C4I\)](https://adafru.it/C4I) module. This module allows you to easily write Python code that reads the luminosity and more from the sensor.

You can use this sensor with any CircuitPython microcontroller board or with a computer that has GPIO and Python thanks to [Adafruit_Blinka, our CircuitPython-for-Python compatibility library \(https://adafru.it/BSN\)](https://adafru.it/BSN).

CircuitPython Microcontroller Wiring

First wire up a TSL2591 to your board exactly as shown on the previous pages for Arduino using an I2C connection. Here's an example of wiring a Feather M0 to the sensor with I2C:




- Board 3V to sensor VIN
- Board GND to sensor GND
- Board SCL to sensor SCL
- Board SDA to sensor SDA

Python Computer Wiring

Since there's *dozens* of Linux computers/boards you can use we will show wiring for Raspberry Pi. For other platforms, [please visit the guide for CircuitPython on Linux to see whether your platform is supported \(https://adafru.it/BSN\)](https://adafru.it/BSN).

Here's the Raspberry Pi wired with I2C:



- Pi 3V3 to sensor VIN
- Pi GND to sensor GND
- Pi SCL to sensor SCK
- Pi SDA to sensor SDA

CircuitPython Installation of TSL2591

Next you'll need to install the [Adafruit CircuitPython TSL2591 \(https://adafru.it/C4I\)](https://adafru.it/C4I) library on your CircuitPython board.

First make sure you are running the [latest version of Adafruit CircuitPython \(https://adafru.it/Amd\)](https://adafru.it/Amd) for your board.

Next you'll need to install the necessary libraries to use the hardware--carefully follow the steps to find and install these libraries from [Adafruit's CircuitPython library bundle \(https://adafru.it/zdx\)](https://adafru.it/zdx). Our introduction guide has [a great page on how to install the library bundle \(https://adafru.it/ABU\)](https://adafru.it/ABU) for both express and non-express boards.

Remember for non-express boards like the, you'll need to manually install the necessary libraries from the bundle:

- `adafruit_tsl2591.mpy`
- `adafruit_bus_device`

Before continuing make sure your board's lib folder or root filesystem has the `adafruit_tsl2591.mpy`, and `adafruit_bus_device` files and folders copied over.

Next [connect to the board's serial REPL \(https://adafru.it/Awz\)](https://adafru.it/Awz) so you are at the CircuitPython >>> prompt.

CircuitPython and Python Usage

To demonstrate the usage of the sensor we'll initialize it and read the luminosity from the board's Python REPL. Run the following code to import the necessary modules and initialize the I2C connection with the sensor:

```
import board
import busio
import adafruit_tsl2591
i2c = busio.I2C(board.SCL, board.SDA)
sensor = adafruit_tsl2591.TSL2591(i2c)
```

Now you're ready to read values from the sensor using any of these properties:

- **lux** - The computed light lux value measured by the sensor.
- **visible** - The visible light level measured by the sensor. This is a 32-bit unsigned value with no units, the higher the number the more visible light.
- **infrared** - The infrared light level measured by the sensor. This is a 16-bit unsigned value with no units, the higher the number the more infrared light.
- **full_spectrum** - The visible & infrared light level measured by the sensor as a single 32-bit value with no units. The higher the number the more full spectrum light.
- **raw_luminosity** - A 2-tuple of raw sensor visible+IR and IR only light levels. Each is a 16-bit value with no units where the higher the value the more light.

```
print('Light: {0}lux'.format(sensor.lux))
print('Visible: {0}'.format(sensor.visible))
print('Infrared: {0}'.format(sensor.infrared))
```

```
>>> print('Light: {0}lux'.format(sensor.lux))
Light: 272.146lux
>>> print('Visible: {0}'.format(sensor.visible))
Visible: 24315761
>>> print('Infrared: {0}'.format(sensor.infrared))
Infrared: 370
>>> █
```

In addition you can read and write a few properties to change how the sensor behaves and is configured:

- **gain** - Set this to a value below to change the gain of the light sensor:
 - `adafruit_tsl2591.GAIN_LOW` - 1x gain
 - `adafruit_tsl2591.GAIN_MED` - 25x gain (the default)
 - `adafruit_tsl2591.GAIN_HIGH` - 428x gain

- `adafruit_tsl2591.GAIN_MAX` - 9876x gain
- **integration_time** - Set the integration time of the sensor to a value of:
 - `adafruit_tsl2591.INTEGRATIONTIME_100MS` - 100ms (the default)
 - `adafruit_tsl2591.INTEGRATIONTIME_200MS` - 200ms
 - `adafruit_tsl2591.INTEGRATIONTIME_300MS` - 300ms
 - `adafruit_tsl2591.INTEGRATIONTIME_400MS` - 400ms
 - `adafruit_tsl2591.INTEGRATIONTIME_500MS` - 500ms
 - `adafruit_tsl2591.INTEGRATIONTIME_600MS` - 600ms

```
sensor.gain = adafruit_tsl2591.GAIN_LOW
sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_200MS
```

```
>>> sensor.gain = adafruit_tsl2591.GAIN_LOW
>>> sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_200MS
>>> print('Light: {0}lux'.format(sensor.lux))
Light: 281.112lux
>>> print('Visible: {0}'.format(sensor.visible))
Visible: 1966237
>>> print('Infrared: {0}'.format(sensor.infrared))
Infrared: 30
>>>
```

That's all there is to using the TSL2591 sensor with CircuitPython!

Below is a complete example that will print the light level every second. Save this as `code.py` on the board and view the output in the REPL.

Full Example Code

```

# Simple demo of the TSL2591 sensor. Will print the detected light value
# every second.
import time

import board
import busio

import adafruit_tsl2591

# Initialize the I2C bus.
i2c = busio.I2C(board.SCL, board.SDA)

# Initialize the sensor.
sensor = adafruit_tsl2591.TSL2591(i2c)

# You can optionally change the gain and integration time:
#sensor.gain = adafruit_tsl2591.GAIN_LOW (1x gain)
#sensor.gain = adafruit_tsl2591.GAIN_MED (25x gain, the default)
#sensor.gain = adafruit_tsl2591.GAIN_HIGH (428x gain)
#sensor.gain = adafruit_tsl2591.GAIN_MAX (9876x gain)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_100MS (100ms, default)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_200MS (200ms)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_300MS (300ms)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_400MS (400ms)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_500MS (500ms)
#sensor.integration_time = adafruit_tsl2591.INTEGRATIONTIME_600MS (600ms)

# Read the total lux, IR, and visible light levels and print it every second.
while True:
    # Read and calculate the light level in lux.
    lux = sensor.lux
    print('Total light: {0}lux'.format(lux))
    # You can also read the raw infrared and visible light levels.
    # These are unsigned, the higher the number the more light of that type.
    # There are no units like lux.
    # Infrared levels range from 0-65535 (16-bit)
    infrared = sensor.infrared
    print('Infrared light: {0}'.format(infrared))
    # Visible-only levels range from 0-2147483647 (32-bit)
    visible = sensor.visible
    print('Visible light: {0}'.format(visible))
    # Full spectrum (visible + IR) also range from 0-2147483647 (32-bit)
    full_spectrum = sensor.full_spectrum
    print('Full spectrum (IR + visible) light: {0}'.format(full_spectrum))
    time.sleep(1.0)

```

Python Docs

[Python Docs \(https://adafru.it/C4n\)](https://adafru.it/C4n)

Downloads


Datasheets & Files

- EagleCAD PCB files on GitHub (<https://adafru.it/pOf>)
- Fritzing object in Adafruit Fritzing library (<https://adafru.it/c7M>)

<https://adafru.it/Foz>


<https://adafru.it/Foz>

Schematic



Layout

(Dimensions are in Inches)



Данный компонент на территории Российской Федерации

Вы можете приобрести в компании MosChip.

Для оперативного оформления запроса Вам необходимо перейти по данной ссылке:

<http://moschip.ru/get-element>

Вы можете разместить у нас заказ для любого Вашего проекта, будь то серийное производство или разработка единичного прибора.

В нашем ассортименте представлены ведущие мировые производители активных и пассивных электронных компонентов.

Нашей специализацией является поставка электронной компонентной базы двойного назначения, продукции таких производителей как XILINX, Intel (ex.ALTERA), Vicor, Microchip, Texas Instruments, Analog Devices, Mini-Circuits, Amphenol, Glenair.

Сотрудничество с глобальными дистрибьюторами электронных компонентов, предоставляет возможность заказывать и получать с международных складов практически любой перечень компонентов в оптимальные для Вас сроки.

На всех этапах разработки и производства наши партнеры могут получить квалифицированную поддержку опытных инженеров.

Система менеджмента качества компании отвечает требованиям в соответствии с ГОСТ Р ИСО 9001, ГОСТ РВ 0015-002 и ЭС РД 009

Офис по работе с юридическими лицами:

105318, г.Москва, ул.Щербаковская д.3, офис 1107, 1118, ДЦ «Щербаковский»

Телефон: +7 495 668-12-70 (многоканальный)

Факс: +7 495 668-12-70 (доб.304)

E-mail: info@moschip.ru

Skype отдела продаж:

moschip.ru

moschip.ru_4

moschip.ru_6

moschip.ru_9